HW 8

1. Stern−Gerlach, spin=1

The program is to compute the eigenvectors of J_x , and their overlaps with the $J_z = \hbar$ state.

From the definition of the ladder operators (Sakurai Eq. 3.5.5), $J_x = (J_+ + J_-)/2$, and $J_\pm$ are (Sakurai Eqs. 3.5.39−41):

```
jp = ℏ {{0, √2, 0}, {0, 0, √2}, {0, 0, 0}};
jm = ℏ {{0, 0, 0}, {√2, 0, 0}, {0, √2, 0}};
jx = (jp + jm) / 2;
```

Solve for the eigenvectors of J_x :

```
jxsol = Eigensystem[jx]
```

$\{\{0, -\hbar, \hbar\}, \{\{-1, 0, 1\}, \{1, -\sqrt{2}, 1\}, \{1, \sqrt{2}, 1\}\}\}$

Extract and normalize:

```
jxup = jxsol[[2, 3]] / Sqrt[jxsol[[2, 3]].jxsol[[2, 3]]]
jxmi = jxsol[[2, 1]] / Sqrt[jxsol[[2, 1]].jxsol[[2, 1]]]
jxdn = jxsol[[2, 2]] / Sqrt[jxsol[[2, 2]].jxsol[[2, 2]]]
```

$\left\{\frac{1}{2}, \frac{1}{\sqrt{2}}, \frac{1}{2}\right\}$

$\left\{-\frac{1}{\sqrt{2}}, 0, \frac{1}{\sqrt{2}}\right\}$

$\left\{\frac{1}{2}, -\frac{1}{\sqrt{2}}, \frac{1}{2}\right\}$

The probability for each state is the square of the overlap with $J_z = \hbar$:

```
jzup = {1, 0, 0};
(jzup.jxup)^2
(jzup.jxmi)^2
(jzup.jxdn)^2
```

$\frac{1}{4}$

$\frac{1}{2}$

$\frac{1}{4}$

As one might expect: the probabilities add up to 1, the $J_x = 0$ probability is highest, and the probabilities of $J_x = \hbar$ and $J_x = -\hbar$ are equal.

N.B.: The J_x eigenstates can also be obtained by applying a $\pi/2$ rotation about y to the J_z eigenstates.

2. Wigner−Eckhart, rank=2, 3

(a) We must calculate

$$\langle \psi_{3,3} \mid Q_0 \mid \psi_{3,3} \rangle = \int d\, r\, r^2\; R^2(r)\, r^2 \int d\, \Omega\, Y_{3,3}(\theta, \phi)^* \; \frac{Q_0}{r^2}\; Y_{3,3}(\theta, \phi)$$

with

$$\frac{Q_0}{r^2} = \frac{1}{2\, r^2}(3\, z^2 - r^2) = \frac{1}{2}(3 \cos^2 \theta - 1).$$

Let us ignore the radial integral (which always factors out as such) and compute the angular integral.

First, define the Q operator as above, and a delayed−evaluation function for the inner product:

```
q0 = 1/2 (3 Cos[θ]^2 - 1);
shavg[m2_, q_, m1_] := Integrate[Conjugate[SphericalHarmonicY[3, m2, θ, ϕ]]
    q SphericalHarmonicY[3, m1, θ, ϕ] Sin[θ], {θ, 0, π}, {ϕ, 0, 2 π}];
```

Ok, compute:

```
q0a33 = shavg[3, q0, 3]
```

$-\frac{1}{3}$

(b) Using the result above, the double−bar inner−product on the RHS of the Wigner−Eckhart theorem is

```
dbavg = q0a33 / ClebschGordan[{3, 3}, {2, 0}, {3, 3}] * √7
```

$-2\sqrt{\frac{7}{15}}$

Then, we can insert this in a general function for any m_1 and m_2 :

```
weavg[m2_, q_, m1_] := ClebschGordan[{3, m1}, {2, q}, {3, m2}] * dbavg / √7;
```

More legibly,

$$\left\langle 3, m_2 \left| (Q^{(2)})_q \right| 3, m_1 \right\rangle = \left(\frac{1}{\sqrt{7}}\right) \langle 3, 2; m_1, q | 3, 2; 3, m_2 \rangle \left(-2\sqrt{\frac{7}{15}}\right)$$

On the RHS, the first factor comes from $1 / \sqrt{2 j_1 + 1}$ where $j_1 = 3$ is the total angular momentum eigenvalue of the ket.

The second factor is the Clebsch−Gordan coefficient of

$$| j_1, m_1 \rangle \otimes | k, q \rangle \Longleftrightarrow | j_2, m_2 \rangle$$

written in the notation of Sakurai; for this problem, $j_1 = j_2 = 3$ for the $l = 3$ ket and bra respectively, and $k = 2$ for the quadropole tensor. For a non−zero result, $m_1 + q = m_2$ and $| j_1 - j_2 | < k < j_1 + j_2$ (which is already satisifed for us as $0 \leq 2 \leq 6$).

The third factor is shown to exist by the Wigner−Eckhart theorem, written $\langle 3 \| Q \| 3 \rangle$ in the notation of Sakurai. It is independent of "magnetic" (i.e., orientational) quantum numbers m_1, m_2 and q, depending only on the "shape" quantum numbers j_1, j_2 and k, (If we wished to include any radial dependence, it would be here as well.)

(c) Let us define our Q operators in similar fashion to part (a):

```
qp1 = -Sqrt[3/2] (Sin[θ] Cos[ϕ] + I Sin[θ] Sin[ϕ]) Cos[θ];
qm2 = Sqrt[3/8] (Sin[θ] Cos[ϕ] - I Sin[θ] Sin[ϕ])^2;
```

And compute the inner products using the ’shavg’ function from part (a):

```
shavg[1, qp1, 0]
shavg[-1, qm2, 1]
shavg[-2, q0, -3]
```

$\frac{\sqrt{2}}{15}$

$-\frac{2\sqrt{\frac{2}{3}}}{5}$

0

Compare to the results using the Wigner−Eckhart function from part (b):

```
weavg[1, 1, 0]
weavg[-1, -2, 1]
weavg[-2, 0, -3]
```

$\frac{\sqrt{2}}{15}$

$-\frac{2\sqrt{\frac{2}{3}}}{5}$

```
ClebschGordan::phy : ThreeJSymbol[{3, -3}, {2, 0}, {3, 2}] is not physical. More…
```

0

Indeed, they are the same.

3. $j_1 = 3/2$, $j_2 = 1$ C−G coeff's [optional]

The first Clebsh−Gordan (C−G) coefficient is by convention:

$C(3/2, 1, 5/2; 3/2, 1, 5/2) = 1$

where we have used the easy−to−type notation $C(j_1, j_2, j; m_1, m_2, m)$.

Let us implement the lowering eigenvalue:

```
lower[j_, m_] = √((j + m) (j - m + 1));
```

We now start stepping downward from the given max z state and solving for the C−G coefficients, using the C−G recursion relation (Sakurai Eq. 3.7.45). First $m = 3/2$:

```
lower[3/2, 3/2] / lower[5/2, 5/2]
lower[1, 1] / lower[5/2, 5/2]
```

$\sqrt{\frac{3}{5}}$

$\sqrt{\frac{2}{5}}$

$C(3/2, 1, 5/2; 1/2, 1, 3/2) = \sqrt{3/5}$
$C(3/2, 1, 5/2; 3/2, 0, 3/2) = \sqrt{2/5}$

$m = 1/2$:

```
√3/5 lower[3/2, 1/2] / lower[5/2, 3/2]
√3/5 lower[1, 1] / lower[5/2, 3/2]

√2/5 lower[3/2, 3/2] / lower[5/2, 3/2]
√2/5 lower[1, 0] / lower[5/2, 3/2]
```

$\sqrt{\frac{3}{10}}$

$\frac{\sqrt{\frac{3}{5}}}{2}$

$\frac{\sqrt{\frac{3}{5}}}{2}$

$\frac{1}{\sqrt{10}}$

$C(3/2, 1, 5/2; -1/2, 1, 1/2) = \sqrt{3/10}$
$C(3/2, 1, 5/2; 1/2, 0, 1/2) = \sqrt{3/5}\,/2 + \sqrt{3/5}\,/2 = \sqrt{3/5}$
$C(3/2, 1, 5/2; 3/2, -1, 1/2) = \sqrt{1/10}$

Note that since the 2nd and 3rd branches end up as $|3/2, 1/2\rangle \otimes |1, 0\rangle$, we just add their contributions.

$m = -1/2$:

```
√3/10 lower[3/2, -1/2] / lower[5/2, 1/2]
√3/10 lower[1, 1] / lower[5/2, 1/2]

√3/5 lower[3/2, 1/2] / lower[5/2, 1/2]
√3/5 lower[1, 0] / lower[5/2, 1/2]

√1/10 lower[3/2, 3/2] / lower[5/2, 1/2]
```

$\frac{1}{\sqrt{10}}$

$\frac{1}{\sqrt{15}}$

$\frac{2}{\sqrt{15}}$

$\sqrt{\frac{2}{15}}$

$\frac{1}{\sqrt{30}}$

$C(3/2, 1, 5/2; -3/2, 1, -1/2) = \sqrt{1/10}$
$C(3/2, 1, 5/2; -1/2, 0, -1/2) = 1\big/\sqrt{15} + 2\big/\sqrt{15} = \sqrt{3/5}$
$C(3/2, 1, 5/2; 1/2, -1, -1/2) = \sqrt{2/15} + \sqrt{1/30} = \sqrt{3/10}$

Note that we omitted the last branch, because it would involve lowering $|1, -1\rangle$.

$m = -3/2$:

```
√(1/10) lower[1, 1] / lower[5/2, -1/2]

√(3/5) lower[3/2, -1/2] / lower[5/2, -1/2]
√(3/5) lower[1, 0] / lower[5/2, -1/2]

√(3/10) lower[3/2, 1/2] / lower[5/2, -1/2]
```

$\frac{1}{2\sqrt{10}}$

$\frac{3}{2\sqrt{10}}$

$\frac{\sqrt{\frac{3}{5}}}{2}$

$\frac{\sqrt{\frac{3}{5}}}{2}$

$C(3/2, 1, 5/2; -3/2, 0, -3/2) = \sqrt{1/10}\,/2 + \sqrt{9/10}\,/2 = \sqrt{2/5}$
$C(3/2, 1, 5/2; -1/2, -1, -3/2) = \sqrt{3/5}\,/2 + \sqrt{3/5}\,/2 = \sqrt{3/5}$

Here we omitted the first and last branches due to annihilation. Also, these results are the same as with $|5/2, 3/2\rangle$, which one would expect by symmetry.

Finally, $m = -5/2$:

```
√(2/5) lower[1, 0] / lower[5/2, -3/2]

√(3/5) lower[3/2, -1/2] / lower[5/2, -3/2]
```

$\frac{2}{5}$

$\frac{3}{5}$

$C(3/2, 1, 5/2; -3/2, -1, -5/2) = 2/5 + 3/5 = 1$

This too is the necessary result, both by convention and symmetry.
We have computed all the non−zero C−G coefficients for $j = 5/2$, as we have covered all the possible ways to create J_z states $-5/2 \le m \le 5/2$ with $|3/2, m_1\rangle \otimes |1, m_2\rangle$. Clearly, one can generalize this process using a tree algorithm.

However, we are not done −− we must also calculate the C−G coefficients for the $j = 1/2, 3/2$ representations. Unfortunately, there is no single extreme z state to exploit in these cases. What are we to do?

One option would be to generate a closed formula for the C−G coeff's for fixed m_1 and m_2 by multiplying the eigenvalues from repeated application of the lowering operator, then using the argument that the C−G coeff's must form an orthogonal transformation between the bases $|j, m\rangle$ and $|j_1, m_1\rangle \otimes |j_2, m_2\rangle$ (see Sakurai, Section 3.7).

However, since we've delineated all the $j = 5/2$ C−G coeff's, we can take a shortcut. We know that the different representations we get when adding angular momentum are separate and irreducible, so eigenstates from different representations with the same z eigenvalue must be orthogonal. Indeed the number of ways to produce a particular z state with $| j_1, m_1 \rangle \otimes | j_2, m_2 \rangle$ is equal to the number irreducible representations.

Knowing this, we can immediately write down the C−G coeff's for the max−z state of the sum representation $j = 3/2$, $|3/2, 3/2\rangle$:

$C(3/2, 1, 3/2; 1/2, 1, 3/2) = -\sqrt{2/5}$
$C(3/2, 1, 3/2; 3/2, 0, 3/2) = \sqrt{3/5}$

Note that we have fixed a sign convention.

Now we proceed in the same fashion as with $j = 5/2$, first $m = 1/2$:

```
-√2/5 lower[3/2, 1/2] / lower[3/2, 3/2]
-√2/5 lower[1, 1] / lower[3/2, 3/2]

√3/5 lower[3/2, 3/2] / lower[3/2, 3/2]
√3/5 lower[1, 0] / lower[3/2, 3/2]
```

$$-2\sqrt{\frac{2}{15}}$$

$$-\frac{2}{\sqrt{15}}$$

$$\sqrt{\frac{3}{5}}$$

$$\sqrt{\frac{2}{5}}$$

$C(3/2, 1, 3/2; -1/2, 1, 1/2) = -\sqrt{8/15}$
$C(3/2, 1, 3/2; 1/2, 0, 1/2) = -2/\sqrt{15} + 3/\sqrt{15} = \sqrt{1/15}$
$C(3/2, 1, 3/2; 3/2, -1, 1/2) = \sqrt{2/5}$

$m = -1/2$:

```
-√8/15 lower[3/2, -1/2] / lower[3/2, 1/2]
-√8/15 lower[1, 1] / lower[3/2, 1/2]

√1/15 lower[3/2, 1/2] / lower[3/2, 1/2]
√1/15 lower[1, 0] / lower[3/2, 1/2]

√2/5 lower[3/2, 3/2] / lower[3/2, 1/2]
```

$-\sqrt{\frac{2}{5}}$

$-\frac{2}{\sqrt{15}}$

$\frac{1}{\sqrt{15}}$

$\frac{1}{\sqrt{30}}$

$\sqrt{\frac{3}{10}}$

$C(3/2, 1, 3/2; -3/2, 1, -1/2) = -\sqrt{2/5}$
$C(3/2, 1, 3/2; -1/2, 0, -1/2) = -2/\sqrt{15} + 1\sqrt{15} = -\sqrt{1/15}$
$C(3/2, 1, 3/2; 1/2, -1, -1/2) = \sqrt{1/30} + \sqrt{9/30} = \sqrt{8/15}$

One might expect this result via symmetry, taking into account the minus sign.

$m = -3/2$:

```
-√2/5 lower[1, 1] / lower[3/2, -1/2]

-√1/15 lower[3/2, -1/2] / lower[3/2, -1/2]
-√1/15 lower[1, 0] / lower[3/2, -1/2]

√8/15 lower[3/2, 1/2] / lower[3/2, -1/2]
```

$-\frac{2}{\sqrt{15}}$

$-\frac{1}{\sqrt{15}}$

$-\frac{\sqrt{\frac{2}{5}}}{3}$

$\frac{4\sqrt{\frac{2}{5}}}{3}$

$C(3/2, 1, 3/2; -3/2, 0, -3/2) = -2/\sqrt{15} - 1/\sqrt{15} = -\sqrt{3/5}$
$C(3/2, 1, 3/2; -1/2, -1, -3/2) = -\sqrt{2/5}/3 + 4\sqrt{2/5}/3 = \sqrt{2/5}$

Again, we could have predicted this result by symmetry; or, calculated it directly by orthogonality to $|5/2, -3/2\rangle$.

Finally, for $j = 1/2$ we can again construct the $m = 1/2$ state, and therefore the C−G coeff's, by orthogonality to both $|5/2, 1/2\rangle$ and $|3/2, 1/2\rangle$. Then we solve

$$\sqrt{3/10}\, x + \sqrt{3/5}\, y + \sqrt{1/10}\, z = 0$$
$$-\sqrt{8/15}\, x + \sqrt{1/15}\, y + \sqrt{2/5}\, z = 0$$

where x, y, z are the C−G coefficients

```
Solve[{√(3/10) x + √(3/5) y + √(1/10) z == 0,
  -√(8/15) x + √(1/15) y + √(2/5) z == 0, x^2 + y^2 + z^2 == 1}, {x, y, z}]
```

$$\left\{\left\{z \to -\frac{1}{\sqrt{2}}, x \to -\frac{1}{\sqrt{6}}, y \to \frac{1}{\sqrt{3}}\right\}, \left\{z \to \frac{1}{\sqrt{2}}, x \to \frac{1}{\sqrt{6}}, y \to -\frac{1}{\sqrt{3}}\right\}\right\}$$

to get

$$C(3/2, 1, 1/2; -1/2, 1, 1/2) = \sqrt{1/6}$$
$$C(3/2, 1, 1/2; 1/2, 0, 1/2) = -\sqrt{1/3}$$
$$C(3/2, 1, 1/2; 3/2, -1, 1/2) = \sqrt{1/2}$$

where we have chosen a sign convention.

We can do the same for $m = -1/2$:

```
Solve[{√(1/10) x + √(3/5) y + √(3/10) z == 0,
  -√(2/5) x - √(1/15) y + √(8/15) z == 0, x^2 + y^2 + z^2 == 1}, {x, y, z}]
```

$$\left\{\left\{z \to -\frac{1}{\sqrt{6}}, x \to -\frac{1}{\sqrt{2}}, y \to \frac{1}{\sqrt{3}}\right\}, \left\{z \to \frac{1}{\sqrt{6}}, x \to \frac{1}{\sqrt{2}}, y \to -\frac{1}{\sqrt{3}}\right\}\right\}$$

to get

$$C(3/2, 1, 1/2; -3/2, 1, -1/2) = \sqrt{1/2}$$
$$C(3/2, 1, 1/2; -1/2, 0, -1/2) = -\sqrt{1/3}$$
$$C(3/2, 1, 1/2; 1/2, -1, -1/2) = \sqrt{1/6}$$

4. Fun with $Y_{l,m}$'s [optional]

(a) The raising operator in the position representation is

$$L_+ = (-i\hbar)\, e^{i\phi}\left(i\,\frac{\partial}{\partial\theta} - \cot\theta\,\frac{\partial}{\partial\phi}\right)$$

(Sakurai Eq. 3.6.13). Applying it to $Y_{2,2}$,

```
SphericalHarmonicY[2, 2, θ, ϕ]
```

$$\frac{1}{4}\, e^{2 i \phi} \sqrt{\frac{15}{2\pi}}\, \text{Sin}[\theta]^2$$

```
(-I ℏ) e^(i ϕ) (I D[%, θ] - Cot[θ] D[%, ϕ])
```

0

(b) The lowering operator is

$$L_- = (-i\,\hbar)\, e^{i\phi} \left(-i \frac{\partial}{\partial\theta} - \cot\theta \frac{\partial}{\partial\phi}\right).$$

We implement both it and its eigenvalue

```
lowy[y_] := (-I ℏ) Exp[-I ϕ] (-I D[y, θ] - Cot[θ] D[y, ϕ]);
lowyv[m_] = √((2 + m) (2 - m + 1)) ℏ;
```

and apply it repeatedly to $Y_{2,2}$:

```
y21 = lowy[SphericalHarmonicY[2, 2, θ, ϕ]] / lowyv[2]
```

$$-\frac{1}{2} e^{i\phi} \sqrt{\frac{15}{2\pi}} \operatorname{Cos}[\theta] \operatorname{Sin}[\theta]$$

```
y21 - SphericalHarmonicY[2, 1, θ, ϕ]
```

0

```
y20 = Simplify[lowy[y21] / lowyv[1]]
```

$$\frac{1}{8} \sqrt{\frac{5}{\pi}} (1 + 3 \operatorname{Cos}[2\theta])$$

```
Simplify[y20 - SphericalHarmonicY[2, 0, θ, ϕ]]
```

0

```
y2m1 = lowy[y20] / lowyv[0]
```

$$\frac{1}{4} e^{-i\phi} \sqrt{\frac{15}{2\pi}} \operatorname{Sin}[2\theta]$$

```
Simplify[y2m1 - SphericalHarmonicY[2, -1, θ, ϕ]]
```

0

```
y2m2 = Simplify[lowy[y2m1]] / lowyv[-1]
```

$$\frac{1}{4} e^{-2i\phi} \sqrt{\frac{15}{2\pi}} \operatorname{Sin}[\theta]^2$$

```
Simplify[y2m2 - SphericalHarmonicY[2, -2, θ, ϕ]]
```

```
0
```

```
lowy[y2m2]
```

```
0
```

(c) Create a plotting function

```
ploty[m_] := ParametricPlot3D[
   Conjugate[SphericalHarmonicY[2, m, θ, ϕ]] SphericalHarmonicY[2, m, θ, ϕ]
    {Sin[θ] Cos[ϕ], Sin[θ] Sin[ϕ], Cos[θ]}, {θ, 0, π}, {ϕ, 0, 2 π}, PlotPoints → 50];
```

and plot $|Y_{2,m}|^2$ for $m = 0, 1, 2$:

```
Table[ploty[m], {m, 0, 2}];
```


-0.1
0.1
-0.05
0.05
0
0.05
0.1
0
-0.05
-0.1
0.4
0.2
0
-0.2
-0.4


0.1
0.05
0
-0.05
-0.1
0.1
0.05
0
-0.05
-0.1
-0.1
-0.05
0
0.05
0.1


0.04
0.02
0
-0.02
-0.04
-0.1
0
0.1
0.1
0
-0.1